
§３ 連続体構造物に対する有限要素解析 

（１）連続体の有限要素分割と変位関数 

▼解析手順 

 連続体構造物に対する有限要素解析も、基本的にはトラス構造の解析と同様の手順で行われる。 

  手順１：要素の剛性方程式：［Ｋ]e{Ｕ}e＝{Ｆ}e を作成する。 

   手順２：構造全系の剛性方程式：［Ｋ]{Ｕ}＝{Ｆ｝を作成する。 

  手順３：外力と変位の境界条件を導入し、剛性方程式（連立一次方程式）を解く。 

  手順４：節点変位｛Ｕ｝から各要素内のひずみ｛ε｝や応力｛σ｝等を計算する。 

▼節点力・節点変位ベクトル 

 連続体構造物を分割表現する要素には種々の形状が考えられるが、ここでは最も簡単な三角形定

ひずみ要素について説明する（図－3.1)。 (ａ)図のように個々の要素は三角形の頂点を節点とし

て互いに連結され、力を及ぼし合っている。(ｂ)図は代表的な要素を示したものであり、節点番号

をｉ，ｊ，ｋとすると、要素の節点力および節点変位ベクトルは次のようになる。 

                          図－3.1  三角形要素 

▼変位関数と形状関数 

 剛性方程式の誘導に際しては、要素内の任意点の変位と節点変位の関係を予め定めておく必要が

あり、このために「変位関数」が定義される。変位関数に用いる最も簡単な関数はｘ，ｙの一次式

であり、この仮定によって要素内ではひずみが一定になるので、定ひずみ要素と呼ばれる。このと

き、｛α｝＝{α1，α2，α3，α4，α5，α6}
Ｔ を未定常数として、要素内の変位ｕ，ｖは 

                  ｕ(ｘ，ｙ)＝ α1 ＋ α2ｘ ＋ α3ｙ 

         ｖ(ｘ，ｙ)＝ α4 ＋ α5ｘ ＋ α6ｙ           (3.2) 

で表される。もちろん、上式以外にｘ，ｙの２次、３次など高次の成分を含む変位関数も考え得る。

次数の高い関数を用いるほど要素内の変位は精度よく表現できるが、その代わり解析が煩雑になる。

上式を行列表示するために、ｕ，ｖをまとめて｛δ｝と置くと 

上式のｘ，ｙに３つの節点の座標を代入すれば｛δ｝は各節点の変位を表す。対応する座標や変位

の成分に節点番号 ｉ,ｊ,ｋ を添字として付け、３つの節点の表示を１つにまとめると 

上式を｛α｝について解き、｛α｝＝[Ｃ]－１{Ｕ}e の形にしてから式(3.3)に代入すると、要素内の

任意点の変位｛δ｝と節点変位{Ｕ}e の関係が定まる。 ［Ｃ］
－１ を求めると 

ｂi＝ｙj－ｙk       (3.5) 

Ａは三角形要素の面積であり、次式より定まる。 

以上から、要素内変位｛δ｝と節点変位{Ｕ}e の関係は次のように表される。 

上の行列［Ｎ］を「形状関数(shape function)」、または要素内部の変位を、節点変位を用いて補

間する意味から「補間関数あるいは内挿関数」と呼ばれる。 

▼形状関数の性質 

  図－3.2 のように三角形ｉｊｋ内に１点Ｐを考え

ると、この三角形は点Ｐに関して面積Ａi, Ａj, Ａk の

３つの三角形に分割される。ただし、節点ｉに対面す

る面積がＡi になるように記号を付ける。点Ｐの位置

によって３つの三角形の面積が変わるので、逆に３つ

の三角形の面積をもって点Ｐの位置、すなわち座標が

定義できる。通常は元の三角形ｉｊｋの面積Ａとの比

をとって無次元化して次のように表す。 

  Ｌi＝Ａi/Ａ, Ｌj＝Ａj/Ａ, Ｌk＝Ａk/Ａ  (3.8)        図－3.2  面積座標 

これを「面積座標」という。上式のＬi などは図の垂 

直距離ｚi，ｈi の比：Ｌi＝ｚi/ｈi で定義されると考えてもよい。面積座標とｘｙ座標の関係は式

(3.6) のような行列式から求めることができる。例えば Ｌi は 

逆に、Ｌi＋Ｌj＋Ｌk＝１ なることを利用して、ｘ，ｙについて解くと 

を得る。 上式は面積座標（Ｌi，Ｌj，Ｌk）を介して節点座標と要素内座標を関係づける式である

が、“座標”を“変位”と読み替えて（ｘ，ｙ）→（ｕ，ｖ）とすれば、節点変位と要素内変位の

関係：式(3.7)を与える。すなわち、式(3.9)で定義した面積座標は形状関数［Ｎ］の成分になる。 

 式(3.9)でｘ、ｙに節点座標を代入して面積座標（Ｌi，Ｌj，Ｌk）の値を調べると 

    ｉ点（ｘ＝ｘi，ｙ＝ｙi） →  Ｌi＝1，Ｌj＝0, Ｌk＝0 

    ｊ点（ｘ＝ｘj，ｙ＝ｙj） →  Ｌi＝0，Ｌj＝1, Ｌk＝0 

    ｋ点（ｘ＝ｘk，ｙ＝ｙk） →  Ｌi＝0，Ｌj＝0, Ｌk＝1 

すなわち、形状関数の成分Ｌiは、ある１つの節点で大きさ“1”、他の節点で“0”となるような関

数であり、例えばＬi を縦軸にとって要素 ijk 内での変化を調べると、図－3.3(ａ)のようになる。

このＬi-分布に節点変位｛δi}＝｛ｕi，ｖi｝を乗じると、要素内の変位は節点ｉで｛δi}、節点ｊ

及びｋでゼロの線形分布になる。同様に、Ｌj-分布に｛δj}、Ｌk-分布に｛δk}を乗じて式(3.11)

の加算を行うと、要素内部の変位｛δ｝は、(ｂ)図のように節点変位の線形結合で表されることに

なる。これが形状関数の物理的意味である。 

                                   図－3.3 形状関数 

 

（２）要素の剛性方程式 

▼力学の基本式 

 §１のバネ系や§２のトラス構造の解析で、構造系の方程式｛Ｆ}＝［Ｋ]{Ｕ｝の誘導に際して

は、以下の３つの基本条件式が使われた。 

  ①力のつり合い式：｛Ｆ｝＝［Ｂ］Ｔ｛Ｓ｝   ｛Ｆ｝：外力（節点力） 

                                             ｛Ｓ｝：内力（バネ力，部材力） 

  ②適合条件式：｛Ｅ｝＝［Ｂ］｛Ｕ｝         ｛Ｕ｝：系の全体的な変形（節点変位） 

                                              ｛Ｅ｝：局部的な変形（伸び量，ひずみ） 

  ③バネの構成式：｛Ｓ｝＝［Ｄ］｛Ｅ｝       ［Ｄ］：力と変形の関係（バネ常数） 

  連続体の適合条件式は、物体内の１点のひずみ｛ε｝と変位｛ｕ，ｖ｝の関係式であり 

上式に｛α｝＝[Ｃ]－１{Ｕ}e を用いると、離散系に関する適合条件式が下式で表される。 

              ｛ε｝＝［Ｈ］[Ｃ]－１{Ｕ}e ＝［Ｂ］{Ｕ}e      (3.13) 

 連続体のバネの方程式（構成式）は、二次元、三次元の応力～ひずみ関係を記述するフック則で

あり、平面応力（二次元）問題では 

上式で Ｅ → Ｅ/(1－ν2)，ν → ν/(1－ν) と置き換えると平面ひずみ問題の［Ｄ］を得る。 

▼仮想仕事の原理 ＝ 力のつり合い式 

 力学の基本式の①と②で、外部的な力｛Ｆ｝と変形｛Ｕ｝の内積を考えると 

      ｛Ｆ｝と｛Ｕ｝の内積：ｗｏ ＝｛Ｆ｝
Ｔ｛Ｕ｝ 

                ＝（［Ｂ］Ｔ｛Ｓ})Ｔ｛Ｕ｝＝｛Ｓ}Ｔ［Ｂ]｛Ｕ｝ 

                ＝｛Ｓ｝Ｔ｛Ｅ｝ 

                ＝ 内部的な力｛Ｓ｝と変形｛Ｅ｝の内積：ｗi 

ｗｏ は外力仕事、ｗｉ は内力仕事と見なせるから、上式は適合条件式を満たす変形とつり合い条

件を満たす力に対して、仮想仕事の原理：ｗｏ＝ｗi が成り立つことを示している。 換言すれば、

適合条件式の仮定の下で、仮想仕事の原理は物体に働く力がつり合うための必要条件であり、つり

合い条件式と同等に扱われるべき性質のものである。 

  二次元の応力～変形問題では（図－3.4） 

の作用の下で、物体の内部には応力｛σ｝が発生し、こ

れに伴ってひずみ｛ε｝が生じる。 ｛σ｝を“内力”、

｛ｐ｝や｛ｂ｝を“外力”と表現すれば、外力の作用の

下で内力が発生し、それらは互いにつり合っている。 

 『仮想仕事の原理』とは“物体内に働く力がつり合い

状態にあれば、物体の仮想的微小変位に対してそれらの 

なす仕事の総和は０である”と表現される。 つり合い      図－3.4 内力と外力 

状態にある物体内の各点において、微小な変位｛dδ｝＝{dｕ，dｖ}Ｔ と、これに伴う微小ひずみ

｛dε｝＝{dεx，dεy，dγxy}
Ｔ が生じたとすると、外力および内力のなす仕事は 

外力仕事：

++∫=

で表される。ただし、∫dＡ は面積積分、∫dＶ は体積積分である。したがって、仮想仕事の原理

は ｗｏ＝ｗｉ または ｗｏ－ｗｉ＝0 となるから、行列表示して次式を得る。 

{ } { } { } { } { } { } dVbddApddVd TTT δδσε ∫+∫=∫     (3.15) 

▼剛性方程式の誘導 

 上式の微小変位｛dδ｝及び微小ひずみ｛dε｝は、対応する微小節点変位を｛dＵ}e と置いたと

き、式(3.5)の {dα}＝[Ｃ]－１{dＵ}e より 

        式(3.3) ：｛dδ｝＝［Ａ］{dα}＝［Ａ］[Ｃ]－１{dＵ}e ＝［Ｎ］{dＵ}e 

     式(3.12)：｛dε｝＝［Ｈ］{dα}＝［Ｈ］[Ｃ]－１{dＵ}e ＝［Ｂ］{dＵ}e 

となる。したがって、これらを式(3.15)に代入すると 

       { } [ ] { } { } [ ] { } { } [ ] { } dVbNdUdApNdUdVBdU TT
e

TT
e

TT
e ∫+∫=∫ σ    

左辺は式(3.14)の構成式：｛σ｝＝[Ｄ]{ε}＝[Ｄ][Ｂ]{Ｕ}e を用いて、更に 

                    { } [ ] [ ] [ ]{ } dVUBDBdU e
TT

e∫  

と表される。 ここで、節点変位{Ｕ}e（および｛dＵ｝e）は座標ｘ，ｙに無関係であるから積分の

外に出すことができ、また上式は任意の仮想微小変位について成り立たねばならないから、両辺か

ら｛dＵ｝e は消去することができる。この結果、最終的に下の表現を得る。 

[ ] { } { } 　～要素の剛性方程式eee FUK =   (3.16) 

この［Ｋ］e および{Ｆ}e が要素の剛性行列および節点力ベクトルであり、{Ｆ}e は更に表面荷重

による項｛Ｆp｝e と物体力による項｛Ｆb｝e に分けられる。定ひずみ要素を仮定すると、剛性行列

［Ｋ］e の積分項［Ｂ］
Ｔ[Ｄ][Ｂ]は要素内で一定値であり、変数ｘ，ｙを含まないから、要素の

体積、面積、厚さを各々 Ｖ，Ａ，ｔ と置くと［Ｋ］e は次式で与えられる。 

[ ] [ ] [ ] [ ] [ ] [ ] [ ] AtBDBVBDBK TT
・・ ==e    (3.17) 

 全系の方程式はトラスの構造解析と同様な方法で要素の方程式を加え合わせて得られる。全節点

数をＮｐとすると［Ｋ］は 2Ｎｐ×2Ｎｐ の対称行列になる。 

▼表面分布荷重｛ｐ｝＝{ｐx，ｐy}
Ｔ の導入 

ｑ 3L 3/2

ｑ 1L 1/2

 {ｐ} による節点力ベクトルは式(3.16)より 

 

                 (3.18) 

で与えられる。∫dＡ は面積積分であるが、二次元

問題では物体表面に沿う線積分になる。しかし、実

際にはこの積分を行うことなく、単純ばりの計算に

よって分布荷重を境界節点に働く等価な集中荷重

に置き換えて表面荷重の効果を取り入れることが

できる（図－3.5）。すなわち、各境界節点が関係す 

る範囲の荷重を集中荷重に置換えて分割配分する。     図－3.5 表面荷重の処理 

▼物体力｛ｂ｝＝{ｂx，ｂy}
Ｔ の導入 

 {ｂ} による節点力ベクトルも上と同形で表され（体積積分∫dＶ である点が異なる） 

ｂx，ｂyを要素内で一定値とし、座標原点を要素の重心にとって積分を実行すると 

構造物の自重による応力・変形問題では、ｂx＝0，ｂy＝－γ であるから、上式によれば要素の重

量（γＡｔ）を３等分したものがｙ方向の節点荷重として３つの節点に分配されると考えればよい。 

▼要素内応力・ひずみの計算 

γAｔ/3

面積A

 方程式の解として節点変位{Ｕ}e を得たのちは、式

(3.13)より要素内のひずみ｛ε｝、式(3.14) より要素

内の応力｛σ｝が定まる。 

    ｛ε｝＝［Ｈ］[Ｃ]－１{Ｕ}e ＝［Ｂ］{Ｕ}e 

    ｛σ｝＝［Ｄ］｛ε｝＝［Ｄ］[Ｂ]{Ｕ}e 

                     (3.20) 

変位関数にｘ，ｙの一次式を仮定すると、応力・ひず

みは要素内で一定値をとる。 

（３）計算の実行 

▼解析上の留意点 

L

H

解析領域

Ｂ

ｐ0

ｐ0

細かく
粗く

Ｂ/2

対称断面

  半無限地盤の表面に一様分布荷重ｐ0 (幅Ｂ) 

が作用する時の地盤内の応力・変形問題を考え 

る(図－3.6)。 この問題では地盤が無限の広が 

りをもっているから、まず(ａ)図のように解析 

領域（Ｈ，Ｌ）の大きさを設定する必要がある。 

Ｈ，Ｌを小さくとると領域境界の拘束の影響が 

応力・変形解に現れるし、大きくとり過ぎると 

要素数を多くしなければならない面がある。適 

当な値は試行計算を行って決定するしかないが、 

通常は Ｌ/Ｂ＞10，Ｈ/Ｂ＞5 程度を考えれば 

よい。 

 次に、この問題は対称形であるから、(ｂ)図 

のように半断面だけを解析領域とすればよいこ 

とが分かる。このとき対称軸の条件は、軸に沿 

うせん断応力τ＝0 であるから、ｘ方向に拘束、 

ｙ方向に自由のロ－ラ－支持を考えればよい。 

また、この種の問題では表面荷重付近で応力が 

急激に変化するので、その変化を精度良く表現 

するために要素分割を細かくする必要がある。      図－3.6 計算上の注意 

 

 

 

 

▼ＦＥＭプログラム 

  （三角形・定ひずみ要素／弾性解析／表面荷重＆自重） 

 

C     ****************************************************************** 

C     *                                                                * 

C     *    ##  feplane.for  ##                2000 / Jan               * 

C     *                                                                * 

C     *    Two-dimensional Elastic Stress-Deformation Analysis         * 

C     *    * Constant Strain Triangular Element                        * 

C     *                                                                * 

C     *     INPUT = 55                                                 * 

C     *    OUTPUT =  6 (print)                                         * 

C     *    OUTPUT = 56 (Displacement)                                  * 

C     *    OUTPUT = 57 (Stress & Strain)                               * 

C     *                                                                * 

C     ****************************************************************** 

C=================== 

C     MAIN PROGRAM 

C=================== 

C 

      character*48 fname0,fname55,fname6,fname56,fname57 

      write(*,'(A)') '***** two-dimensional elastic analysis *****' 

      write(*,'(A)') '      (feplane.for) triangular elemwnt' 

      write(*,'(A)') '    Input data file ===>> ./data/test.txt' 

 read (*,'(A)')  fname0 

 open(1,file=fname0) 

 read(1,*) fname55,fname6,fname56,fname57     ← 入出力ファイル名の読み込み 

      close(1) 

C 

 open(55,file=fname55) 

 open( 6,file=fname6)                         ← 入出力ファイルの open 

 open(56,file=fname56) 

 open(57,file=fname57) 

C 

      CALL INPUT          ... デ－タ入力 

      CALL EBMAT          ... 行列［Ｂ］の作成  （プログラムでは［EB］） 

      CALL DMAMAT         ... 弾性行列［Ｄ］の作成（プログラムでは［DMA］） 

      CALL STIFF          ... 剛性行列［Ｋ］の作成（プログラムでは［ST］） 

      CALL PLOAD          ... 荷重ベクトル｛Ｐ｝の作成 

      CALL DISPL          ... 境界条件の導入／連立一次方程式の計算 

      CALL STRESS         ... 応力・ひずみの計算 

C 

      close(55) 

   close(6)                                       ← 入出力ファイルの close 

   close(56) 

      close(57) 

C 

      STOP 

      END 

C 

      SUBROUTINE INPUT 

C========================================== 

C      READING AND PRINTING OF INPUT DATA                     ... デ－タ入出力 

C========================================== 

C     1 NELEM    ** TOTAL NUMBER OF ELEMENTS                  ... 要素数 

C       NPOIN    ** TOTAL NUMBER OF NODAL POINTS              ... 節点数 

C       NBLK     ** NUMBER OF BLOCKS                          ... 材質ブロック数 

C       NPLANE   ** NPLANE=1 ... PLANE STRAIN                 ... 平面ひずみ／平面応力 

C                   NPLANE=2 ... PLANE STRESS                      

C       NLOAD    ** NLOAD=1 .... BODY FORCE ONLY              ... 荷重条件 

C                   NLOAD=2 .... SURFACE LOAD ONLY                ＊自重と表面荷重 

C                   NLOAD=3 .... BODY FORCE AND SURFACE LOAD 

C     2 NOD(J,I) ** THREE NODAL POINT NUMBERS OF ELEMENT      ... 要素を囲む３節点 

C     3 XY(J,I)  ** X,Y-COORDINATES OF NODAL POINT            ... 節点のｘｙ座標 

C     4 KODE(I)  ** KODE(I)=0 .. X,Y-FREE                     ... 変位の境界条件 

C                   KODE(I)=1 .. X-FIX AND Y-FREE 

C                   KODE(I)=2 .. X-FREE AND Y-FIX                           

C                   KODE(I)=3 .. X,Y-FIX 

C     5 NEB(I)   ** BLOCK NUMBER OF ELEMENT                   ... 要素の材質番号 

C     6 DIMX,DIMY * MULTIPLIER OF X,Y-COORDINATES             ... ｘｙ座標倍率 

C     7 BELA(K)  ** YOUNG'S MODULUS OF EACH BLOCK             ... ヤング率Ｅ（材質ごと） 

C       BYNT(K)  ** POISSON'S RATIO OF EACH BLOCK                 ポアソン比ν 

C       BGAM(K)  ** UNIT WEIGHT     OF EACH BLOCK                 単位体積重量γ 

C     ..... FOR SURFACE LOAD (NLOAD.GE.2) ..... 

C     8 NTPL     ** TOTAL NUMBER OF NODAL POINTS ON SURFACE LOAD  ... 表面荷重の作用点数 

C     9 NPL(K)   ** NUMBER OF NODAL POINT ON SURFACE LOAD             荷重作用点の番号 

C    10 PXY(K,J) ** X,Y-SURFACE LOAD                                  ｘｙの荷重値 

C 

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      COMMON /BL1/ NELEM,NPOIN,NEQ,NB,NBLK,NPLANE,NLOAD 

      COMMON /BL2/ NOD(3,NE0),XY(2,NP0),NEB(NE0),KODE(NP0) 

      COMMON /BL3/ EB(6,NE0),DMA(3,NE0),AREA(NE0) 

      COMMON /BL4/ U(NP0),V(NP0),EPP(3,NE0),SSS(3,NE0) 

      COMMON /BL5/ BELA(9),BYNT(9),BGAM(9) 

      COMMON /BL6/ NTPL,NPL(30),PXY(30,2) 

C 

      READ(55,*) NELEM,NPOIN,NBLK,NPLANE,NLOAD                ... 自由書式デ－タ入力 

      READ(55,*) (N,(NOD(J,I),J=1,3),I=1,NELEM)                  

      READ(55,*) (N,XY(1,I),XY(2,I),I=1,NPOIN)                ... NOD は反時計回りに読む 

      READ(55,*) (KODE(I),I=1,NPOIN) 

      READ(55,*) (NEB(I),I=1,NELEM) 

      READ(55,*) DIMX,DIMY 

      DO 1 I=1,NBLK 

    1 READ(55,*) BELA(I),BYNT(I),BGAM(I)                      ... 単位を統一すること 

      IF(NLOAD.LT.2) GO TO 5                                      （kN-cm） 

      READ(55,*) NTPL 

      READ(55,*) (NPL(I),I=1,NTPL) 

      READ(55,*) (PXY(I,1),PXY(I,2),I=1,NTPL) 

    5 CONTINUE 

C 

C     BAND WIDTH *** NB                                       ... バンド幅（NB）の決定 

C 

      NB=0 

      DO 20 I=1,NELEM 

      DO 20 J=1,3 

      J1=J+1 

      IF(J1.EQ.4) J1=1 

      NODD=ABS(NOD(J,I)-NOD(J1,I)) 

      IF (NODD.GT.NB) NB=NODD 

   20 CONTINUE 

      NB=2*(NB+1) 

C                                                                            

      WRITE(6,2000) NB                                         

      WRITE(6,2010) NELEM,NPOIN,NBLK,NPLANE,NLOAD             ... 入力デ－タの出力 

      WRITE(6,2020) (I,NEB(I),(NOD(J,I),J=1,3),I=1,NELEM) 

      WRITE(6,2030) (I,KODE(I),XY(1,I),XY(2,I),I=1,NPOIN) 

      WRITE(6,2040) DIMX,DIMY 

      DO 2 I=1,NBLK 

    2 WRITE(6,2050) I,BELA(I),BYNT(I),BGAM(I) 

      IF(NLOAD.LT.2) GO TO 6 

      WRITE(6,2060) (NPL(I),PXY(I,1),PXY(I,2),I=1,NTPL) 

    6 CONTINUE 

C 

 2000 FORMAT(1H ,40('*')/1H ,'***  INPUT DATA  ***', 

     *       ' .....  NB =',I3/1H ,40('*')) 

 2010 FORMAT(/1H ,'NELEM * NPOIN * NBLK * NPLANE * NLOAD' 

     *      //1H ,I5,I8,I7,I9,I8) 

 2020 FORMAT(/1H ,'ELEMENT *** NEB *** THREE NODES'// 

     *        1H ,2('  NO. BLOCK   I    J    K  ',3X) 

     *       /(1H ,2(I4,I6,3I5,5X))) 

 2030 FORMAT(/1H ,'POINT *** KODE *** COORDINATES(X,Y)' 

     *      //1H ,2('  NO. KODE     X        Y  ',5X) 

     *      /(1H ,2(I4,I5,2F9.2,5X))) 

 2040 FORMAT(/1H ,'DIMX * DIMY ***'//1H ,2F10.2 

     *      //1H ,'BLOCK DATA ***'/) 

 2050 FORMAT( 1H ,2X,'BLK',I3,5X,'BELA=',F10.3,5X,'BYNT=',F7.3, 

     *           5X,'BGAM=',F10.4) 

 2060 FORMAT(/1H ,'SURFACE LOAD ***'/1H ,/(1H ,2X,'NP.',I5,5X, 

     *            'PX=',F10.3,5X,'PY=',F10.3)) 

C 

      NEQ=2*NPOIN                                             ... 初期値の設定 

      DO 10 I=1,NPOIN                                             変数のゼロクリア 

      XY(1,I)=XY(1,I)*DIMX 

   10 XY(2,I)=XY(2,I)*DIMY 

      DO 12 I=1,NELEM 

      DO 12 J=1,3 

      SSS(J,I)=0.0 

   12 EPP(J,I)=0.0 

      DO 14 I=1,NPOIN 

      U(I)=0.0 

   14 V(I)=0.0 

C 

      RETURN                                                 

      END 

C 

      SUBROUTINE EBMAT                                                      

C===================== 

C     FORM EB-MATRIX                                          ... 行列［Ｂ］の作成 

C=====================                                           

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      COMMON /BL1/ NELEM,NPOIN,NEQ,NB,NBLK,NPLANE,NLOAD 

      COMMON /BL2/ NOD(3,NE0),XY(2,NP0),NEB(NE0),KODE(NP0) 

      COMMON /BL3/ EB(6,NE0),DMA(3,NE0),AREA(NE0) 

      DIMENSION XI(3),YI(3) 

C 

      DO 20 N=1,NELEM 

      DO 21 I=1,3 

      JJ=NOD(I,N) 

      XI(I)=XY(1,JJ) 

   21 YI(I)=XY(2,JJ) 

      B1=YI(2)-YI(3) 

      B2=YI(3)-YI(1) 

      B3=YI(1)-YI(2) 

      C1=XI(3)-XI(2) 

      C2=XI(1)-XI(3) 

      C3=XI(2)-XI(1) 

      AA=(C1*YI(1)+C2*YI(2)+C3*YI(3))*0.5 

      AREA(N)=AA                                              ... AREA(N)＝三角形要素 

      A2=1.0/(2.0*AA)                                                      の面積 

      EB(1,N)=B1*A2 

      EB(2,N)=B2*A2 

－FEM 3.11－ 



      EB(3,N)=B3*A2 

      EB(4,N)=C1*A2 

      EB(5,N)=C2*A2 

      EB(6,N)=C3*A2 

   20 CONTINUE 

C 

      RETURN 

      END 

C 

      SUBROUTINE DMAMAT 

C====================== 

C     FORM DMA-MATRIX                                         ... 弾性行列［Ｄ］の作成 

C====================== 

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      COMMON /BL1/ NELEM,NPOIN,NEQ,NB,NBLK,NPLANE,NLOAD 

      COMMON /BL2/ NOD(3,NE0),XY(2,NP0),NEB(NE0),KODE(NP0) 

      COMMON /BL3/ EB(6,NE0),DMA(3,NE0),AREA(NE0) 

      COMMON /BL5/ BELA(9),BYNT(9),BGAM(9)                     

      DIMENSION BDMA(9,3) 

C 

      DO 40 I=1,NBLK                                                        

      EE=BELA(I) 

      YN=BYNT(I) 

      YN1=1.0+YN 

      YN2=1.0-YN 

      YN3=1.0-2.0*YN 

      IF(NPLANE.EQ.1) GO TO 10 

      YN4=YN1*YN2 

      BDMA(I,1)=EE/YN4 

      BDMA(I,2)=EE*YN/YN4 

      GO TO 15 

   10 YN4=YN1*YN3 

      BDMA(I,1)=EE*YN2/YN4                                    ... 材質毎の弾性係数 

      BDMA(I,2)=EE*YN/YN4                                         （BDMA） 

   15 BDMA(I,3)=EE/(2.0*YN1) 

   40 CONTINUE 

C 

      DO 50 N=1,NELEM 

      NBL=NEB(N)                                              ... 要素毎の弾性係数 

      DO 60 J=1,3                                                 （DMA） 

   60 DMA(J,N)=BDMA(NBL,J) 

   50 CONTINUE 

C 

      RETURN 

      END 

C 

      SUBROUTINE STIFF 

C================================= 

C     FORM STIFFNESS MATRIX (ST)                              ... 要素剛性行列の作成 

C=================================                                全体剛性行列への組込 

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      COMMON /BL1/ NELEM,NPOIN,NEQ,NB,NBLK,NPLANE,NLOAD 

      COMMON /BL2/ NOD(3,NE0),XY(2,NP0),NEB(NE0),KODE(NP0) 

      COMMON /BL3/ EB(6,NE0),DMA(3,NE0),AREA(NE0) 

      COMMON /BL8/ ST(NQ0,200),P(NQ0) 

      DIMENSION D(3,3),BA(3,6),DBA(3,6),C(6,6),LM(6) 

C 

      DO 10 I=1,NB 

      DO 10 J=1,NEQ 

   10 ST(J,I)=0.0 

      DO 1 I=1,3 

      DO 2 K=1,3 

    2 D(I,K)=0.0                                               

      DO 1 J=1,6 

    1 BA(I,J)=0.0 

C                                                                           

      DO 100 N=1,NELEM 

      AA=AREA(N) 

      BA(1,1)=EB(1,N)                                         ... [BA]＝[Ｂ] 

      BA(1,3)=EB(2,N) 

      BA(1,5)=EB(3,N) 

      BA(2,2)=EB(4,N) 

      BA(2,4)=EB(5,N) 

      BA(2,6)=EB(6,N) 

      BA(3,1)=BA(2,2) 

      BA(3,3)=BA(2,4) 

      BA(3,5)=BA(2,6) 

      BA(3,2)=BA(1,1) 

      BA(3,4)=BA(1,3) 

      BA(3,6)=BA(1,5) 

      D(1,1)=DMA(1,N)                                         ... 弾性行列[Ｄ] 

      D(1,2)=DMA(2,N) 

      D(3,3)=DMA(3,N) 

      D(2,1)=D(1,2) 

      D(2,2)=D(1,1) 

      DO 3 I=1,3 

      DO 3 J=1,6 

      DBA(I,J)=0.0 

      DO 3 K=1,3 

      DBA(I,J)=DBA(I,J)+D(I,K)*BA(K,J)                        ... [DBA]＝[Ｄ][Ｂ] 

    3 CONTINUE 

      DO 4 I=1,6 

      DO 4 J=1,6 

      C(I,J)=0.0 

      DO 4 K=1,3 

      C(I,J)=C(I,J)+BA(K,I)*DBA(K,J)*AA                       ... 要素剛性行列 

    4 CONTINUE                                                    [Ｃ]＝[Ｂ]Ｔ[Ｄ][Ｂ]Ａ 

C 

      LM(2)=2*NOD(1,N)                                        ... 要素剛性行列と全体 

      LM(1)=LM(2)-1                                               剛性行列の位置関係 

      LM(4)=2*NOD(2,N) 

      LM(3)=LM(4)-1 

      LM(6)=2*NOD(3,N) 

      LM(5)=LM(6)-1 

      DO 200 I=1,6 

      DO 200 J=1,6                                            ... 全体行列[Ｋ]＝[ST] 

 

      II=LM(I)                                                    への組込み 

      JJ=LM(J)                                                  [Ｋ]が対称かつ帯行列 

      IF(II-JJ) 210,210,200                                       であることを利用 

  210 IJ=JJ-II+1                                                            

      ST(II,IJ)=ST(II,IJ)+C(I,J)                               

  200 CONTINUE                                                 

  100 CONTINUE 

C 

      RETURN 

      END 

C 

      SUBROUTINE PLOAD 

C=========================== 

C     FORM LOAD VECTOR (P)                                    ... 荷重ベクトル{Ｐ}の作成 

C=========================== 

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      COMMON /BL1/ NELEM,NPOIN,NEQ,NB,NBLK,NPLANE,NLOAD 

      COMMON /BL2/ NOD(3,NE0),XY(2,NP0),NEB(NE0),KODE(NP0) 

      COMMON /BL3/ EB(6,NE0),DMA(3,NE0),AREA(NE0) 

      COMMON /BL5/ BELA(9),BYNT(9),BGAM(9) 

      COMMON /BL6/ NTPL,NPL(30),PXY(30,2) 

      COMMON /BL8/ ST(NQ0,200),P(NQ0) 

C 

      DO 300 I=1,NEQ 

  300 P(I)=0.0 

      IF(NLOAD.EQ.2) GO TO 310 

C 

C     BODY FORCE                                              ... 自重項 

C 

      DO 100 N=1,NELEM 

      NBL=NEB(N) 

      WW=BGAM(NBL)*AREA(N)/3.0                                 

      DO 110 I=1,3                                            ... ｙ方向の荷重成分として 

      JJ=2*NOD(I,N)                                               γＡ/3 ずつ分配 

  110 P(JJ)=P(JJ)+WW 

  100 CONTINUE 

      IF(NLOAD.EQ.1) RETURN 

  310 CONTINUE 

C 

C     SURFACE LOAD                                            ... 表面荷重項 

C 

      DO 200 I=1,NTPL 

      NN=2*NPL(I) 

      P(NN-1)=P(NN-1)+PXY(I,1)                                ... デ－タ入力した PXY 

  200 P(NN  )=P(NN  )+PXY(I,2)                                    をｘｙ方向に加算 

C                                         

      RETURN 

      END 

C                                                                            

      SUBROUTINE DISPL 

C============================================= 

C     BOUNDARY CONDITIONS AND SOLVE EQUATION                  ... 変位の境界条件の導入 

C============================================= 

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      COMMON /BL1/ NELEM,NPOIN,NEQ,NB,NBLK,NPLANE,NLOAD 

      COMMON /BL2/ NOD(3,NE0),XY(2,NP0),NEB(NE0),KODE(NP0) 

      COMMON /BL4/ U(NP0),V(NP0),EPP(3,NE0),SSS(3,NE0) 

      COMMON /BL8/ ST(NQ0,200),P(NQ0) 

C 

C     BOUNDARY CONDITION 

C 

      DO 500 N=1,NPOIN 

      KO=KODE(N) 

      NN=2*N 

      IF(KO-1) 500,150,125 

  125 IF(KO-3) 250,150,500 

  150 NNI=NN-1 

      UV=U(N) 

      GO TO 200 

  250 KO=KO-1 

      NNI=NN 

      UV=V(N) 

  200 IF(NNI.GE.NB) GO TO 355 

      M=NNI 

      GO TO 360 

  355 M=NB 

  360 DO 400 II=1,M 

      K=NNI-(II-1) 

      P(K)=P(K)-ST(K,II)*UV 

  400 ST(K,II)=0.0                                            ... [Ｋ]が対称かつ帯行列 

      DO 425 II=2,NB                                              である特性を考慮 

      K=NNI+(II-1) 

      IF(K-NEQ) 450,450,425 

  450 P(K)=P(K)-ST(NNI,II)*UV 

  425 ST(NNI,II)=0.0 

      ST(NNI,1)=1.0 

      P(NNI)=UV 

      IF(KO.EQ.3) GO TO 250 

  500 CONTINUE 

C 

C     SOLVE EQUATION                                          ... 対称・帯行列の[Ｋ]の 

C                                                               連立一次方程式を解く 

      CALL SYMEQ (ST,P,NEQ,NB)                                    （GAUSS の消去法） 

C                                                                          

      DO 600 N=1,NPOIN 

      NN=2*N 

      U(N)=U(N)+P(NN-1)                                       ... 変位値Ｕ，Ｖ 

  600 V(N)=V(N)+P(NN  ) 

      WRITE(6,3000) (I,U(I),V(I),I=1,NPOIN)                   ... 変位の出力（印刷出力） 

      WRITE(56,*)   (I,U(I),V(I),I=1,NPOIN)                   ... 変位の出力（ﾌｧｲﾙ出力） 

 3000 FORMAT(//1H ,'***** DISPLACEMENTS *****'                    

     1       //1H ,2('POINT',5X,'U(CM)',8X,'V(CM)',2X,'* ') 

     2       /(1H ,2(I4,2X,1PE10.3,' , ',1PE10.3,' * '))) 

C 

      RETURN 

      END 

C 

      SUBROUTINE STRESS 

C======================================= 

C     CALCULATION OF STRESS AND STRAIN                        ... 応力・ひずみ計算 

C======================================= 

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      COMMON /BL1/ NELEM,NPOIN,NEQ,NB,NBLK,NPLANE,NLOAD 

      COMMON /BL2/ NOD(3,NE0),XY(2,NP0),NEB(NE0),KODE(NP0) 

      COMMON /BL3/ EB(6,NE0),DMA(3,NE0),AREA(NE0) 

      COMMON /BL4/ U(NP0),V(NP0),EPP(3,NE0),SSS(3,NE0) 

      COMMON /BL8/ ST(NQ0,200),P(NQ0) 

      DIMENSION DIS(6),EPS(3),STS(3) 

C 

      PEI=3.141593 

      WRITE(6,4000) 

 4000 FORMAT(///1H ,'STRESS AND STRAIN *****'//) 

      DO 700 N=1,NELEM 

      DO 710 I=1,3 

      II=2*I 

      JJ=2*NOD(I,N) 

      DIS(II-1)=P(JJ-1) 

  710 DIS(II  )=P(JJ  ) 

      EPS(1)=EB(1,N)*DIS(1)+EB(2,N)*DIS(3)+EB(3,N)*DIS(5)     ... {ε}＝[Ｂ]{Ｕ}e 

      EPS(2)=EB(4,N)*DIS(2)+EB(5,N)*DIS(4)+EB(6,N)*DIS(6) 

      EPS(3)=EB(1,N)*DIS(2)+EB(2,N)*DIS(4)+EB(3,N)*DIS(6) 

     1+EB(4,N)*DIS(1)+EB(5,N)*DIS(3)+EB(6,N)*DIS(5) 

      STS(1)=DMA(1,N)*EPS(1)+DMA(2,N)*EPS(2)                  ... {σ}＝[Ｄ]{ε} 

      STS(2)=DMA(2,N)*EPS(1)+DMA(1,N)*EPS(2)                   

      STS(3)=DMA(3,N)*EPS(3) 

      DO 720 I=1,3                                             

      SSS(I,N)=SSS(I,N)+STS(I) 

  720 EPP(I,N)=EPP(I,N)+EPS(I) 

      DO 730 I=1,3                                                           

      STS(I)=SSS(I,N) 

  730 EPS(I)=EPP(I,N) 

C 

      CALL PRNCPL (PEI,STS(1),STS(2),STS(3),   ST1,ST2,ALPHA,TMX)   ... 主応力の計算 

      CALL PRNCPL (PEI,EPS(1),EPS(2),EPS(3)*.5,EP1,EP2,ALEPS,GMX)   ... 主ひずみの計算 

      GMX=2.0*GMX                                                   ... γｍａｘ＝GMX 

C 

      WRITE(6,4001) N,TMX,GMX 

      WRITE(6,4002) (STS(I),I=1,3),ST1,ST2,ALPHA 

      WRITE(6,4003) (EPS(I),I=1,3),EP1,EP2,ALEPS 

      WRITE(57,*) N,(STS(I),I=1,3),(EPS(I),I=1,3),GMX       ...応力・ひずみ値のﾌｧｲﾙ出力 

 4001 FORMAT(1H ,'ELEM',I5,12X,'TAU=',E14.5,3X,'EPG=',E14.5) 

 4002 FORMAT(1H ,'STX=',E14.6,3X,'STY=',E14.6,3X,'TXY=',E14.6,5X,'ST1=', 

     *E14.6,3X,'ST2=',E14.6,5X,'ALPHA=',F6.2) 

 4003 FORMAT(1H ,'EPX=',E14.6,3X,'EPY=',E14.6,3X,'EXY=',E14.6,5X,'EP1=', 

     *E14.6,3X,'EP2=',E14.6,5X,'ALEPS=',F6.2) 

  700 CONTINUE 

C 

      RETURN 

      END 

      SUBROUTINE PRNCPL (PEI,S1,S2,S3,P1,P2,AN,XM)       ... 主応力・主ひずみの計算 

      SUM=(S1+S2)*0.5 

      DIF=S1-S2 

      DD=DIF*DIF+4.0*S3*S3 

      XM=SQRT(DD)*0.5 

      P1=SUM+XM                                 

      P2=SUM-XM 

      AN=PEI*0.25 

      IF(DIF.NE.0.0) ANG=0.5*ATAN(2.0*S3/DIF) 

      IF(S1.LT.S2)   ANG=ANG+PEI*0.5  

      AN=ANG*180.0/PEI 

      RETURN 

      END 

C 

      SUBROUTINE SYMEQ (ST,P,NEQ,NB) 

C*********************************** 

C     SOLVE SYMMETRIC EQUATION                                ... GAUSS の消去法 

C*********************************** 

      PARAMETER (NE0=1500,NP0=1000,NQ0=2000) 

      DIMENSION ST(NQ0,200),P(NQ0),SST(200) 

      L=0                                                      

      MNB=NB 

      DO 1 N=1,NEQ                                            ... 前進消去 

      ST1=ST(N,1)                                

      IF(ST1.EQ.0.0) GO TO 1 

      P(N)=P(N)/ST1 

      NNEQ=N-NEQ 

      IF(NNEQ) 15,20,15 

   15 IF(NB.GE.(1-NNEQ)) MNB=1-NNEQ 

      DO 2 I=1,NB 

    2 SST(I)=ST(N,I) 

      DO 3 J=2,MNB 

      IF(ST(N,J)) 35,3,35 

   35 ST(N,J)=ST(N,J)/ST1 

      JJ=N+J-1 

      NNB=MNB-J+1 

      DO 4 I=1,NNB 

      IN=I+J-1 

    4 ST(JJ,I)=ST(JJ,I)-SST(IN)*ST(N,J) 

      P(JJ)=P(JJ)-SST(J)*P(N) 

    3 CONTINUE 

    1 CONTINUE 

C 

   20 NNB=NB-1                                                ... 後退代入 

      L=L+1 

      N=N-1 

      IF(N) 50,50,55 

   55 IF(L-NNB) 60,60,65 

   60 LB=L 

      GO TO 70 

   65 LB=NNB 

   70 DO 8 J=1,LB 

      JJ=N+J 

    8 P(N)=P(N)-ST(N,J+1)*P(JJ) 

      GO TO 20 

C 

   50 RETURN 

      END 

▼計算例：半無限地盤上の一様載荷に対する地表面変位と地盤内の応力解を求める。 

     地盤材料：Ｅ＝20,000 kN/m2＝20MPa，ν＝0.3，γ＝0.0 

          表面荷重：ｑ＝50kN/m2＝50kPa 

＊なお、この計算は四角形定ひずみ要素弾性解析プログラム（fqplane.for）で行う。 

有限要素分割 

 

 

＊用意するデ－タファイル：strip4.txt（入出力ファイル名デ－タ） 

  './data/strip4.dat'     入力デ－タ 

  './data/strip4.lst'    変位・応力解の印刷出力用 

  './data/strip4dis.dat'   変位解のファイル出力用 

'./data/strip4str.dat'   応力解のファイル出力用 

▼入力デ－タ（strip4.dat）                                                     

※このデ－タは四角形要素用プログラム（fqplane.for）用に作成したものである。三角形要素に

対しては、NOD の読み込みで、3,4 番面の節点値を同一にすることで四角形と区別する。 

 

  90  67  1  1  2 

1  1 14 15 15    2  1 15  2  2    3  2 15  3  3    4  3 15 16 16   

   5  3 16 17 17    6  3 17  4  4    7  4 17  5  5    8  5 17 18 18    

   9  5 18 19 19   10  5 19  6  6   11  6 19  7  7   12  7 19 20 20    

  13  7 20 21 21   14  7 21  8  8   15  8 21  9  9   16 14 22 15 15    

  17 15 22 23 23   18 15 23 24 24   19 15 24 16 16   20 16 24 17 17    

  21 17 24 25 25   22 17 25 26 26   23 17 26 18 18   24 18 26 19 19    

  25 19 26 27 27   26 19 27 28 28   27 19 28 20 20   28 20 28 21 21    

  29 22 34 23 23   30 23 34 24 24   31 24 35 25 25   32 25 35 26 26    

  33 26 36 27 27   34 27 36 28 28   35 22 37 34 34   36 34 37 38 38    

  37 24 34 38 38   38 24 38 35 35   39 35 38 39 39   40 26 35 39 39            

  41 26 39 36 36   42 36 39 40 40   43 28 36 40 40   44 21 28 29 29              

  45  9 21 29 29   46  9 29 30 30   47  9 30 10 10   48 10 30 11 11                

  49 11 30 31 31   50 28 40 29 29   51 11 31 32 32   52 11 32 12 12         

  53 12 32 13 13   54 13 32 33 33   55 29 40 41 41   56 29 41 30 30       

  57 30 41 31 31   58 31 41 42 42   59 31 42 32 32   60 32 42 43 43   

  61 32 43 33 33   62 37 44 38 38   63 38 44 45 45   64 38 45 39 39       

  65 39 45 40 40   66 40 45 46 46   67 40 46 41 41   68 41 46 47 47                 

  69 41 47 42 42   70 42 47 48 48   71 42 48 49 49   72 42 49 43 43       

  73 44 50 45 45   74 45 50 51 51   75 45 51 52 52   76 45 52 46 46          

  77 46 52 47 47   78 47 52 53 53   79 47 53 54 48   80 48 54 55 49              

  81 50 56 57 51   82 51 57 58 52   83 52 58 59 53   84 53 59 60 54        

  85 54 60 61 55   86 56 62 63 57   87 57 63 64 58   88 58 64 65 59                 

  89 59 65 66 60   90 60 66 67 61 

   1  0.0 20.0   2  1.0 20.0   3  2.0 20.0   4  3.0 20.0             

   5  4.0 20.0   6  5.0 20.0   7  6.0 20.0   8  7.0 20.0     

   9  8.0 20.0  10 10.0 20.0  11 12.0 20.0  12 16.0 20.0       

  13 20.0 20.0  14  0.0 19.0  15  1.0 19.0  16  2.0 19.0     

  17  3.0 19.0  18  4.0 19.0  19  5.0 19.0  20  6.0 19.0     

  21  7.0 19.0  22  0.0 18.0  23  1.0 18.0  24  2.0 18.0     

  25  3.0 18.0  26  4.0 18.0  27  5.0 18.0  28  6.0 18.0       

  29  8.0 18.0  30 10.0 18.0  31 12.0 18.0  32 16.0 18.0       

  33 20.0 18.0  34  1.0 17.0  35  3.0 17.0  36  5.0 17.0        

  37  0.0 16.0  38  2.0 16.0  39  4.0 16.0  40  6.0 16.0      

  41 10.0 16.0  42 15.0 16.0  43 20.0 16.0  44  0.0 13.0          

  45  4.0 13.0  46  7.0 13.0  47 10.0 13.0  48 15.0 13.0       

  49 20.0 13.0  50  0.0 10.0  51  4.0 10.0  52  8.0 10.0     

  53 12.0 10.0  54 16.0 10.0  55 20.0 10.0  56  0.0  6.0    

  57  4.0  6.0  58  8.0  6.0  59 12.0  6.0  60 16.0  6.0    

  61 20.0  6.0  62  0.0  0.0  63  4.0  0.0  64  8.0  0.0    

  65 12.0  0.0  66 16.0  0.0  67 20.0  0.0 

 1 0 0 0 0 0 0 0 0 0   0 0 3 1 0 0 0 0 0 0   0 1 0 0 0 0 0 0 0 0 

 0 0 3 0 0 0 1 0 0 0   0 0 3 1 0 0 0 0 3 1   0 0 0 0 3 1 0 0 0 0 

 3 3 3 3 3 3 3              

 1 1 1 1 1 1 1 1 1 1   1 1 1 1 1 1 1 1 1 1   1 1 1 1 1 1 1 1 1 1 

 1 1 1 1 1 1 1 1 1 1   1 1 1 1 1 1 1 1 1 1   1 1 1 1 1 1 1 1 1 1  

 1 1 1 1 1 1 1 1 1 1   1 1 1 1 1 1 1 1 1 1   1 1 1 1 1 1 1 1 1 1  

 1.   1. 

 1 20000.   0.3    0. 

   3 

   1   2   3 

 0.  -25.    0.  -50.    0.   -25. 

▼計算結果と表示                                                              

○鉛直応力の等値線 

 σz＝10kPa
（σz/ｑ＝0.2）

 鉛直応力（σz）の等値線理論解
約3.0Ｂ

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

主応力図

 

 

 

 

 

 

 

   ○主応力分布 

○変位・変形図 

 

 

 約1.

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

○変位ベク 

 トル図 

 

 

 

 

 

 

 

 

 

沈下量
5cm

変位・変形図

変位ベクトル図

2cm

